
Computer Graphics – Hands-on 
Creating Photo-Realistic Images with POV-Ray (I) 

 

  

In this hands-on session we will experiment with a ray tracing tool called POV-Ray that can be 
used to make photo-realistic images. As you work with POV-Ray you can begin to observe the 
differences between the graphics rendered by POV-Ray and the graphics rendered by Blender 
and OpenGL. Using POV-Ray will also give you experience with vectors, transformations, 
positioning the camera, and lighting in different context.  
 
Note: These notes have been adapted from the POV-Ray tutorial 
 

http://wiki.povray.org/content/Documentation:Tutorial_Table_of_Contents 
 
 

 
 
 
 

http://wiki.povray.org/content/Documentation:Tutorial_Table_of_Contents

2.2.1  Our First Image 
 
We will create the scene file for a simple picture. Since ray-tracers thrive on spheres, that is what 
we will render first.  
 
2.2.1.1  Understanding POV-Ray's Coordinate System 
 
First, we have to tell POV-Ray the position and orientation of our camera. To do this, we use 3D 
coordinates. The usual coordinate system for POV-Ray has the positive y-axis pointing up, the 
positive x-axis pointing to the right, and the positive z-axis pointing into the screen as follows:  
  
 
 
 
 
 
 
 
This kind of coordinate system is called a left-handed coordinate system. If we use our left 
hand's fingers we can easily see why it is called left-handed. We just point our thumb in the 
direction of the positive x-axis (to the right), the index finger in the direction of the positive y-
axis (straight up) and the middle finger in the positive z-axis direction (forward). We can only do 
this with our left hand. If we had used our right hand we would not have been able to point the 
middle finger in the correct direction.  
 
The left hand can also be used to determine rotation directions. To do this we must perform the 
famous "Computer Graphics Aerobics" exercise. We hold up our left hand and point our thumb 
in the positive direction of the axis of rotation. Our fingers will curl in the positive direction of 
rotation. Similarly if we point our thumb in the negative direction of the axis our fingers will curl 
in the negative direction of rotation.  
 
In the above illustration, the left hand is curling around the x-axis. The thumb points in the 
positive x direction and the fingers curl over in the positive rotation direction.  
 
2.2.1.2  Adding Standard Include Files 
 
Use File->New File to create a new text file in the POV-Ray editor. Use File->Save As and save 
the (empty) file to your directory with the name demo.pov. This is a POV-ray scene file where 
you can write a description of the scene you want to render. Type in the text below. The input is 
case sensitive, so we have to be sure to get capital and lowercase letters correct.  
 

#include "colors.inc"       
#include "stones.inc"    

The include files contain pre-defined scene elements. The first include statement reads in 
definitions for various useful colors. The second include statement reads in a collection of stone 
textures.  
 
2.2.1.3  Adding a Camera 
 
The camera statement describes where and how the camera sees the scene. It specifies the x, y 
and z coordinates to indicate the position of the camera and what point of the scene the camera is 
pointing at (similar to lookat in OpenGL). The camera coordinates are specified using a three-
element vector. In POV-Ray, a vector is given by three numeric values placed between a pair of 
angle brackets and separated by commas. Add the following camera statement to your scene file.  
 
  camera { 
    location <0, 2, -3> 
    look_at  <0, 1,  2> 
  } 

 
Briefly, location <0,2,-3> places the camera up two units and back three units from the 
center of the ray-tracing universe, which is at <0,0,0>. By default +z is into the screen and -z is 
back out of the screen.  
The statement look_at <0,1,2> rotates the camera to point at point <0,1,2>, one unit up from 
the origin and two units away from the origin. This makes it five units in front of and one unit 
lower than the camera. The look_at point should be the center of attention of our image.  
 
2.2.1.4  Describing an Object 
 
Now that the camera is set up to record the scene, let us place a yellow sphere into the scene. 
Add the following lines of code to your scene file:  
 
  sphere { 
    <0, 1, 2>, 2 
    texture { 
      pigment { color Yellow } 
    } 
  } 
 
The first vector specifies the center of the sphere. In this example the x coordinate is zero, so the 
sphere is centered left and right. The y and z coordinates are 1 and 2 (so the sphere is five units 
in front of the camera, which is at z = -3). The second integer specifies the radius (two units) of 
the sphere. (Note that the center vector and the radius value are separated by a comma.) Since the 
radius is half the width of a sphere, the sphere is four units wide.  
 
2.2.1.5  Adding Texture to an Object 
 
In the code above for the sphere, after we defined the location and size of the sphere, we 
described the appearance of the surface using a texture statement. Texture blocks describe the 
color, bumpiness and finish properties of an object. In this example we have specified the color 
only. This is the minimum we must do. All other texture options (except the color) will use 
default values.  

The color we define is the way we want an object to look if fully illuminated. If we were painting 
a picture of a sphere we would use dark shades of a color to indicate the shadowed side, and 
bright shades on the illuminated side. However, ray-tracing takes care of that for you. We only 
need to pick the basic color inherent in the object and POV-Ray brightens or darkens it 
depending on the lighting in the scene. This is why the parameter is called pigment – it defines 
a property (basic color) of an object, not the appearance of the object. 
  
Many types of color patterns are available for use in a pigment statement. The keyword color 
specifies that the whole object is to be one solid color rather than some pattern of colors. We can 
use one of the color identifiers previously defined in the standard include file colors.inc.  
If no standard color is available for our needs, we may define our own color by using the color 
keyword followed by red, green, and blue keywords specifying the amount of red, green and 
blue to be mixed. For example a nice shade of pink can be specified using 
 

  color red 1.0 green 0.8 blue 0.8 
 
Note: the International - rather than American - form "colour" is also acceptable and may be 
used anywhere that "color" may be used.  
 
The values after each keyword should be in the range from 0.0 to 1.0. Any of the three 
components not specified will default to 0. A shortcut notation may also be used. The following 
shortcut statement produces the same shade of pink:  
 

  color rgb <1.0, 0.8, 0.8> 
 
In many cases the color keyword is superfluous, so the shortest way to specify the pink color 
is:  

 rgb <1.0, 0.8, 0.8> 
 
2.2.1.6  Defining a Light Source 
 
One more detail is needed for our scene. We need a light source. Until we create one, there is no 
light in this virtual world. Thus, add the line  
 
  light_source {<2, 4, -3> color White} 
 
to the scene file to get your first complete POV-Ray scene file as shown below: 
 

    #include "colors.inc" 
  camera { 
    location <0, 2, -3> 
    look_at  <0, 1,  2> 
  } 
  sphere { 
    <0, 1, 2>, 2 
    texture { 
      pigment { color Yellow } 
    } 
  } 
  light_source { <2, 4, -3> color White} 

The vector in the light_source statement specifies the location of the light as two units to our 
right, four units above the origin and three units back from the origin. The light source is an 
invisible tiny point that emits light. It has no physical shape, so no texture is needed.  
 
That's it! Now render the scene using Render->Start Rendering from the top menu. POV-ray will 
create an image of the scene described in your file and display it for you.  
 
2.2.2  Basic Shapes 
 
So far we have just used the sphere shape. There are many other types of shapes that can be 
rendered by POV-Ray. The following sections will describe how to use some of the more simple 
objects as a replacement for the sphere above.  
 
2.2.2.1  Box Object 
 
The box is one of the most common objects used in a scene. Try this example by commenting 
out the sphere description, typing in the box description below in your file demo.pov, and then 
rendering the scene.   
 

  box { 
    <-1, 0,   -1>,   // Near lower left corner 
    < 1, 0.5,  3>    // Far upper right corner 
    texture { 
      T_Stone25      // Pre-defined from stones.inc 
      scale 4        // Scale the texture pattern by the  
    // same amount in all directions 
    } 
    rotate y*20      // Equivalent to "rotate <0,20,0>", or  
    // rotate 20 degrees about the Y axis 
} 

 
In this example we can see that a box is defined by the 3D coordinates of its opposite corners. 
The first vector generally holds the minimum x-, y- and z-coordinates, and the second vector 
holds the maximum x-, y- and z-values; however, any two opposite corners may be used. Box 
objects can only be defined parallel to the axes of the world coordinate system. We can later 
rotate them to any angle.  
 
Note: We can perform simple math on values and vectors. In the rotate parameter we multiplied 
the vector identifier y by 20. This is the same as <0,1,0>*20 or <0,20,0>.  
 
2.2.2.2  Cone Object 
 
Here is another example showing how to use a cone:  
 

  cone { 
    <0, 1, 0>, 0.3    // Center and radius of one end 
    <1, 2, 3>, 1.0    // Center and radius of other end 
    texture { T_Stone25 scale 4 } 
  } 

The cone shape is defined by the center and radius of each end. In this example, one end is at 
location <0,1,0> and has a radius of 0.3, while the other end is centered at <1,2,3> with a radius 
of 1. If we want the cone to come to a sharp point we must use radius=0. The solid end caps are 
parallel to each other and perpendicular to the cone axis. If we want an open cone with no end 
caps we have to add the keyword open after the 2nd radius like this:  
 

  cone { 
    <0, 1, 0>, 0.3    // Center and radius of one end 
    <1, 2, 3>, 1.0    // Center and radius of other end 
    open              // Removes end caps 
    texture { T_Stone25 scale 4 } 
  } 

 
2.2.2.3  Cylinder Object 
 
We may also define a cylinder like this:  
 

  cylinder { 
    <0, 1, 0>,     // Center of one end 
    <1, 2, 3>,     // Center of other end 
    0.5            // Radius 
    open           // Remove end caps 
    texture { T_Stone25 scale 4 } 
  } 

 
Try this now by commenting out the box and adding this description of a cylinder and then 
rendering the scene. 
 
2.2.2.4  Plane Object 
 
Let us try out a computer graphics standard, "The Checkered Floor". First, modify your scene so 
that it only includes the original sphere. Then add the following to your scene: 
 

  plane { <0, 1, 0>, -1 
    pigment { 
      checker color Red, color Blue 
    } 
  } 

 
The object defined here is an infinite plane. The vector <0,1,0> is the surface normal of the plane 
(i.e. if we were standing on the surface, the normal points straight up). The number afterward is 
the distance that the plane is displaced by along the normal from the origin -- in this case, the 
floor is placed at y=-1 so that the sphere at y=1, radius=2, is resting on it.  
 
Note: Even though there is no texture statement, there is an implied texture here. We might find 
that continually typing statements that are nested like texture {pigment} can get tiresome, so 
POV-Ray lets us leave out the texture statement under many circumstances. In general we only 
need the texture block surrounding a texture identifier (like the T_Stone25 example above), or 
when creating layered textures (which are covered later). This pigment uses the checker color 
pattern and specifies that the two colors red and blue should be used.  

Because the vectors <1,0,0>, <0,1,0> and <0,0,1> are used frequently, POV-Ray has three built-
in vector identifiers x, y and z respectively that can be used as a shorthand. Thus the plane could 
be defined as:  

  plane { y, -1 
    pigment { ... } 

      } 
Note: We do not use angle brackets around vector identifiers.  
 
Looking at the floor, we notice that the ball casts a shadow on the floor. Shadows are calculated 
very accurately by the ray-tracer, which creates precise, sharp shadows. In the real world, 
penumbral or "soft" shadows are often seen. Later we will learn how to use extended light 
sources to soften the shadows.  

1  
Hit Ctrl-Alt-Print Screen to capture a copy of the sphere and checkered floor using to capture the 
image. Attach the source  code and the image to a Word document to be handed in. 
 
2.2.2.5  Torus Object 
 
A torus can be thought of as a donut or an inner-tube. It is a shape that is vastly useful in many 
kinds of graphics, so POV-Ray has adopted this 4th order quartic polynomial as a primitive 
shape. The syntax for a torus is so simple that it makes it very easy to work with, once we learn 
what its major/minor radius values are. Instead of a lecture on the subject, let us create a torus 
and experiment with it.  Create a new file called torusdemo.pov and edit it as follows:  
 

  #include "colors.inc" 
  camera { 
    location <0, .1, -25> 
    look_at 0    
  } 
  background { color Gray50 } // to make the torus easy to see 
  light_source { <300, 300, -1000> White } 
  torus { 
    4, 1              // major and minor radius 
    rotate -90*x      // so we can see it from the top 
    pigment { Green } 
  } 

 
Render the scene to see the image. Well, it is a donut all right. Let's try changing the major and 
minor radius values and see what happens. Change them as follows and render the scene again:  
 

  torus { 5, .25      // major and minor radius 
 
That looks more like a hula-hoop! Now try these values:  
  

 torus { 3.5, 2.5    // major and minor radius 
 
Whoa! A donut with a serious weight problem! With such a simple syntax, there is not much else 
we can do to a torus besides change its texture... or is there? Let's see...  

Tori are very useful objects in computer graphics. Let's try a little experiment. We make a 
difference of a torus and a box. Replace the torus in your scene file with the following:  
 

 difference { 
    torus { 
      4, 1 
      rotate x*-90  // so we can see it from the top 
    } 
    box { <-5, -5, -1>, <5, 0, 1> } 
    pigment { Green } 
  } 

 
Render the scene. Interesting... a half-torus. Now we add another half-torus flipped the other 
way. To simplify things, let's declare the original half-torus and the necessary transformations so 
we can use them again. Modify your half torus description so that it looks like the following:   
 

  #declare Half_Torus = difference { 
    torus { 
      4, 1 
      rotate -90*x  // so we can see it from the top 
    } 
    box { <-5, -5, -1>, <5, 0, 1> } 
    pigment { Green } 
  } 

 
And add the following two variables at the end of your file: 
 

  #declare Flip_It_Over = 180*x; 
  #declare Torus_Translate = 8;  // twice the major radius 

 
Now add the code below to create a union of two Half_Torus objects:  
 

  union { 
    object { Half_Torus } 
    object { Half_Torus 
      rotate Flip_It_Over 
      translate Torus_Translate*x 
    } 
  } 

Render the scene. This makes an S-shaped object, but we cannot quite see the whole thing from 
our present camera. So let's move the S-shaped object along the +z-direction and rotate it about 
the +y-axis so we can see more of it. Add these transformations to your S-shaped object, as 
shown below:    
 

  union { 
    object { Half_Torus } 
    object { Half_Torus 
      rotate Flip_It_Over 
      translate Torus_Translate*x 
    } 
    rotate y*45 
    translate z*20 
 } 

 
Render the scene. Neat! But we want to model something useful, something that we might see in 
real life. How about a chain? Thinking about it for a moment, we realize that a single link of a 
chain can be easily modeled using two half tori and two cylinders, as shown below: 
 
Create a new file called chain.pov. Cut and paste 
the contents from torusdemo.pov into chain.pov. 
Modify the code so that the object it describes is not 
S-shaped, but instead a single link of a chain like the 
one shown to the right. 
 
Once you have it working, print a picture of the 
rendered scene and attach it (along with your code) 
to the Word document to be handed in. 
 
But whoever heard of a green chain? We would 
rather use a nice metallic color instead. To fix this, remove all pigment blocks in your scene 
description. Then add near the top of your file the declaration for a golden texture below.  

8 
2 

 
  #declare Chain_Gold = texture { 
    pigment { BrightGold } 
    finish { 
      ambient .1 
      diffuse .4 
      reflection .25 
      specular 1 
      metallic 
    } 
  } 

 
To make your chain link appear with this texture, set the texture of your chain link object at the 
bottom of its block, similar to that shown below: 

 
union { 
  /* description of chain link is here: 
   * two half tori connected by two cylindrical segments */ 
   texture { Chain_Gold }  /* last thing in the union block */ 
} 

Render the link now. Much better!  
 
Assign your link object the name Link using the declare keyword. Follow the syntax used to 
declare the Half_Torus object. 
 
Now create a new object that is the union of four links, 
similar on the right. Note that you may want to change 
your camera’s position so you can get a good view of 
the entire chain.  

3  
Once you have a good view, print a copy of your code 
and your rendered image, and attach them to the Word 
document to be handed in.  
 
 
 
 
 

4  
To finish up the topic of basic shapes, create a scene like the one below consisting of a martini 
glass resting on a black and white checkered surface. Make sure your light source is positioned 
so that it casts approximately the same shadow. Once you have it working correctly, save a copy 
of your code and your rendered image to hand-in. 

2.2.3  CSG Objects 

Constructive Solid Geometry, or CSG, is a powerful tool for combining primitive objects to 
create more complex objects, as discussed in this handout.  

2.2.3.1  What is CSG? 

CSG stands for Constructive Solid Geometry. POV-Ray allows us to construct complex solids by 
combining primitive shapes in four different ways: 
 
1. union: two or more shapes are added together.  

2. intersection: two or more shapes are combined to make a new shape that consists of the 
area common to both shapes.  

3. difference: an initial shape has all subsequent shapes subtracted from it.  

4. merge: like a union, but the surfaces inside the union are removed (useful in transparent 
CSG objects).  

 
We will deal with each of these in detail in the next few sections. CSG objects can be extremely 
complex. They can be deeply nested. In other words, there can be unions of differences or 
intersections of merges or differences of intersections or even unions of intersections of 
differences of merges... ad infinitum. CSG objects are (almost always) finite objects and thus 
respond to auto-bounding and can be transformed like any other POV primitive shape.  

2.2.3.2  CSG Union 

Let's try making a simple union. Create a file called csgdemo.pov and edit it as follows:  
 

  #include "colors.inc" 
  camera { 
    location <0, 1, -10> 
    look_at 0 
    angle 36 
  } 
  light_source { <500, 500, -1000> White } 
  plane { y, -1.5 
    pigment { checker Green White } 
  } 

 
Add two spheres, each translated 0.5 units along the x-axis in each direction. Color one blue and 
the other red: 
 

  sphere { <0, 0, 0>, 1 
    pigment { Blue } 
    translate -0.5*x 
  } 
  sphere { <0, 0, 0>, 1 
    pigment { Red } 
    translate 0.5*x 
  } 

Render this image and observe the result. Now place a union block around the two spheres. This 
will create a single CSG union out of the two objects.  
 

  union{ 
    sphere { <0, 0, 0>, 1 
      pigment { Blue } 
      translate -0.5*x 
    } 
    sphere { <0, 0, 0>, 1 
      pigment { Red } 
      translate 0.5*x 
    } 
  } 

 
Trace the file again. The union will appear no different from what each sphere looked like on its 
own, but now we can give the entire union a single pigment. Do that now.  
 

  union{ 
    sphere { <0, 0, 0>, 1 
      translate -0.5*x 
    } 
    sphere { <0, 0, 0>, 1 
      translate 0.5*x 
    } 
    pigment { Red } 
  } 

5  
Trace the file again. Now transform the entire union as a whole by scaling it by a factor of 0.25 
along the Y axis. As we can see, the object has changed dramatically. Save a copy of your image 
to turn in. Label it “Union of two spheres that has been flattened”.  
 
There are many advantages of assigning attributes (e.g. pigments and/or transformations) to the 
entire CSG object, instead of assigning them to each individual component: 
 It is much easier to change the appearance of a CSG object with many components. The 

change involves altering a single line of code, as opposed to altering one line for each 
component.  

 It can save memory, because the attributes are stored once and then referenced by all 
components of the CSG object. Assigning attributes to each of the N components would 
mean N times more memory storage.  

2.2.3.3  CSG Intersection 

Now let's use these same spheres to illustrate the intersection CSG object. Change the word 
union to intersection and remove the scale statement:  
 

  intersection { 
    sphere { <0, 0, 0>, 1 
      translate -0.5*x 
    } 
    sphere { <0, 0, 0>, 1 
      translate 0.5*x 
    } 
    pigment { Red } 
  } 

Render the image and you will see a lens-shaped object instead of the two spheres. This is 
because an intersection consists of the area shared by both shapes, in this case the lens-shaped 
area where the two spheres overlap. We will use this lens-shaped object to also demonstrate 
differences.  

2.2.3.4  CSG Difference 

Add to your intersection a rotate command to rotate the lens-shaped intersection about the y-
axis so that the broad side is facing the camera, as depicted below: 
 

  
 
 
 
 
 
 
 

Now add a blue cylinder to the scene that passes right through the middle of the lens and sticks 
out on both sides, as shown below:  

Once you have that looking right, place a difference block around both the lens-shaped 
intersection and the cylinder, similar to this:  
 

  difference { 
         intersection { 
               sphere { <0, 0, 0>, 1 
                     translate -0.5*x 
                } 
               sphere { <0, 0, 0>, 1 
                      translate 0.5*x 
                } 
               pigment { Red } 
               rotate 90*y 
        } 
        
       /* 
        * your blue cylinder code is here 
        */ 
  } 

 
Render the file again and see the lens-shaped intersection with a neat hole in the middle where 
the cylinder was. The cylinder has been subtracted from the intersection. Note that the pigment 
of the cylinder causes the surface of the hole to be colored blue. If we eliminate this pigment the 
surface of the hole will be black, the default color. 
  
OK, let's get a little wilder now. First, declare our perforated lens object and give it a name. 
Second, eliminate all pigments in the declared object, because we will want a single declaration 
for the final union instead. With these changes, your code should look something like this:  
 

  #declare Lens_With_Hole = difference { 
    intersection { 
      sphere { <0, 0, 0>, 1 
        translate -0.5*x 
      } 
      sphere { <0, 0, 0>, 1 
        translate 0.5*x 
      } 
      rotate 90*y 
    } 
    // cylinder code here without the pigment 
  } 

 
 
Now add code that creates a union of these Lens_With_Hole objects, like that shown below on 
the left. Make the color of the entire union red.  

An interesting object. But let's try something more. Let's make the object partially transparent by 
adding some filter to the pigment block.  Replace the line of code that sets the pigment of the 
entire union to red with the following:  
 
    pigment { Red filter .5 } 
 
Render the file again. This looks pretty good... only... we can see parts of each of the lens objects 
inside the union! (You have to look closely!) This is not good.  

2.2.3.5  CSG Merge 

This brings us to the fourth kind of CSG object, the merge. Merges are the same as unions, but 
the geometry of the objects inside the merge is not traced. This should eliminate the problem 
with our object. Try it now by replacing the word union in your file with the word merge. 

6  
Sure enough, it does! Save a copy of your image to turn in, and label it “Four Copies of 
Lens_With_Hole, merged together”. 

2.2.3.6  CSG Pitfalls 

There is a severe pitfall in the CSG code that we have to be aware of.  

2.2.3.6.1  Co-incident Surfaces 

POV-Ray uses inside/outside tests to determine the points at which a ray intersects a CSG object. 
A problem arises when the surfaces of two different shapes coincide because there is no way 
(due to the computer's floating-point accuracy) to tell whether a point on the coincident surface 
belongs to one shape or the other.  

Modify your code so that it displays only the following object which uses a cylinder to cut a hole 
in a larger box.  
 
  
 
 
  
 
 

  difference { 
    box { -1, 1 pigment { Red } }  
    cylinder { -z, z, 0.5 pigment { Green } } 
  } 

 
Note: that the vectors -1 and 1 in the box definition expand to <-1,-1,-1> and <1,1,1> 
respectively.  
 
Render this object now and notice the green speckles where the hole is suppose to be. Above is a 
diagram showing the configuration. The front and back face of the box and the front and back 
face of the cylinder rest in exactly the same planes, which causes the problem. Which one is in 
front? For some pixels the calculations will determine that the cylinder is in front, and for some 
pixels they will determine the box is. Thus sometimes you get the red box face and sometimes 
you get the green cylinder face appearing.   
 

7  
This problem can be avoided by increasing the size of the cylinder to get rid of the coincidence 
surfaces. Do this now by increasing the length of the cylinder by .001 on both ends.  Save a 
picture of the corrected image to turn in.  
 
In general we have to make the subtracted object a little bit larger in a CSG difference. We just 
have to look for coincident surfaces and increase the subtracted object appropriately to get rid of 
those surfaces.  The same problem occurs in CSG intersections and is also avoided by scaling 
some of the involved objects. 

